Produce a graphical representation of a Sierpinski triangle of order N in any orientation.

An example of Sierpinski's triangle (order = 8) looks like this: 




  [ $ "turtleduck.qky" loadfile ] now!
 
  [ 1 & ]                 is odd        (   n --> b )
 
  [ 4 times 
      [ 2dup walk 
        1 4 turn ] 
    2drop ]               is square     ( n/d -->   )
 
  [ dup
    witheach
      [ odd if
          [ ' [ 0 0 0 ] fill
              [ 2 1 square ] ]
        2 1 fly ]
     size -2 * 1 fly
     1 4 turn 
     2 1 fly
     -1 4 turn ]          is showline   (   [ -->   )
 
  [ [] 0 rot 0 join
    witheach
     [ tuck +
       rot join swap ] 
    drop ]                is nextline   (   [ --> [ )
 
  [ ' [ 1 ] swap
    bit 
    1 - times
      [ dup showline
        nextline ]
    showline ]            is sierpinski (   n -->   )
 
  turtle
  5 8 turn 
  400 1 fly
  3 8 turn 
  8 sierpinski